

Il s'agit de rédaction, et non de passation. Après avoir examiné un aphasique, et hors de la présence du patient, ce programme permet d'accourcir le temps de rédaction du bilan.

Non étalonné, purement descriptif, il propose pour chacune des 20 épreuves classiques du bilan d'aphasie, un choix de 6 options possibles, plusieurs d'entre-elles pouvant être associées. Si aucune ne convient, la CASE VIDE permet d'entrer la réponse désirée, en toutes lettres. Si l'épreuve n'était pas justifiée, SANS OBJET permet de l'occulter, lors de l'impression.

Ce programme n'est pas capable de tirer les conclusions du bilan, mais il offre 3 lignes pour les entrer, en toutes lettres.

Créé et utilisé quotidiennement depuis 2 ans sur un Apple II + de 36 K de mémoire, avec lecteur de disquettes, attelé à une imprimante CENTRONICS 150, ce programme occupe 29 K. La copie du LISTING demande 3 heures ; la rédaction et l'impression en 3 exemplaires d'un bilan prennent moins de 10 minutes. Après l'impression, le bilan est irrémédiablement effacé.

A. DELAFOSSE
Phoniatre
Service de Neurologie C
Hôpital B - CHU
59037 LILLE Cédex

RÉDACTION D'UN BILAN D'APHASIE SUR APPLE II +

par Annette DELAFOSSE

Listing

```

10 DIM A$(20,9),B$(23),GA$(23),RI$(23)
20 FOR I = 0 TO 20
30 FOR J = 1 TO 9
40 READ A$(I,J)
50 NEXT J
55 NEXT I
60 DATA "REUSSITE","ECHEC","PRESENCE DE","ABSENCE DE","RARES","NOMBREUX-
    SES",":",",","(",")"
61 DATA "LATERALISATION","DROITE","GAUCHE","EXCLUSIVE","PREFERENTIELLE","
    PRESUMEE","AMBIDEXTRE","A PRECISER","CASE VIDE"
62 DATA "LANGAGE SPONTANE","PHRASES > 7 MOTS","NUL","LOGORRHEE","REDUIT",
    "STEREOTYPIE","PERSEVERATIONS","MANQUES DU MOT","CASE VIDE"
63 DATA ".....","BIEN ARTICULE","ANARTHRIE","JARGON","PARAPHASI
    ES","SEMANTIQUES","PHONEMIQUES","SANS OBJET","CASE VIDE"
64 DATA ".....","EUSYNTAXIQUE","AGRAMMATIQUE","SYNTAXE RUDIMENT
    AIRE","SYNTAXE ECONOMIQUE","DYSPROSODIE","INSUFFISANCE ARTICULATOIRE"
    ,"SANS OBJET","CASE VIDE"
65 DATA "LANGAGE AUTOMATIQUE","PARTIELLE","SUR COMPTAGE","SUR JOURS","SUR
    MOIS","SUR CHANSONS","PERSEVERATIONS","ANARTHRIE","CASE VIDE"
66 DATA "REPETITION DE MOTS","ANARTHRIE","LEGERE","SEVERE","STEREOTYPIE",

```

"PERSEVERATIONS", "PARAPHASIES PHONEMIQVES", "SUR MOTS LONGS", "CASE VIDE
E"

67 DATA "REPETITION DE PHRASES", "ANARTHRIE", "PHRASE COURTE", "PHRASE LONGU
E", "ACCOURCIE", "COMPLIQUEE", "AGRAMMATIQUE", "SANS OBJET", "CASE VIDE"

68 DATA "PRAXIES BUCCO FACIALES", "PARTIELLE", "SUR IMITATION", "SUR ORDRE",
"PERSEVERATIONS", "AJOUTS", "ERREURS", "SANS OBJET", "CASE VIDE"

69 DATA "DENOMINATION", "REUSSITES", "EHECS", "PAR FACILITATION", "PERSEVERA
TIONS", "RETARDS", "FORMULES D'EHEC", "PAR L'USAGE", "PARTIES D'OBJETS"

70 DATA ".....", "PARAPHASIES", "SEMANTIQUES", "PHONEMIQUES", "VERBALES
MORPHOLOGIQUES", "FANTASTIQUES", "NEOLOGISMES", "PARAGNOSIES", "PERIPHRA
SES"

71 DATA "COMPREHENSION ORALE", "PERSEVERE", "LABORIEUSE", "AUX ORDRES SIMPLE
S", "AU CHOIX LIMITE", "AU CHOIX MULTIPLE", "AUX 3 PAPIERS", "A LA ZEME C
ONSIGNE", "CASE VIDE"

72 DATA "LECTURE A HAUTE VOIX", "ABOLIE", "ANALYTIQUE", "PARALEXIES", "SUR SU
BSTANTIFS", "SUR MOTS GRAMMATICAUX", "JARGONALEXIE", "ANARTHRIE", "CASE V
IDE"

73 DATA "DESIGNATION", "AU HASARD", "PARTIELLE", "MEILLEURE", "DES MOTS", "DES
LETTRES", "DES SYLLABES", "SANS OBJET", "CASE VIDE"

74 DATA "COMPREHENSION ECRITE", "RAPIDE", "LABORIEUSE", "MEME AVEC AIDE", "AU
X MOTS-IMAGES", "AUX ORDRES SIMPLES", "AUX ORDRES COMPLEXES", "AU RESUME
DE TEXTE", "CASE VIDE"

75 DATA "GRAPHISME", "CORRECT", "ABOLI", "DEFICITAIRE", "APRAXIQUE", "ALTERE",
"DE LA MAIN DROITE", "DE LA MAIN GAUCHE", "CASE VIDE"

76 DATA "ORTHOGRAPHE", "FAUTES D'USAGE", "JARGONAGRAPHIE", "PARAGRAPHIES", "P
ROCHES DU MOT-CIBLE", "AGRAMMATISME", "DYSSYNTAXIE", "EN EPELANT", "CASE
VIDE"

77 DATA "GERSTMANN", "ASOMATOGNOSIE VERBALE", "AGNOSIE DIGITALE", "DES 3 DOI
GTS MEDIANS", "CONFUSION DROITE-GAUCHE", "AUX ORDRES CROISES", "ACALCULI
E", "SANS OBJET", "CASE VIDE"

78 DATA "SIGNES FRONTAUX", "ASPONTANEITE", "FLUENCE BASSE", "PERSEVERATIONS"
"DIGRESSIONS", "INDIFFERENCE", "MAINTIEN DES ATTITUDES", "SANS OBJET", "
CASE VIDE"

79 DATA "SIGNES SPATIAUX", "NEGLIGENCE DROITE DANS", "NEGLIGENCE GAUCHE DA
NS", "L'APPREHENSION DE L'ESPACE", "LA LECTURE", "L'ECRITURE", "LE DESSIN
", "SANS OBJET", "CASE VIDE"

80 DATA "AGNOSIE VISUELLE", "DES IMAGES", "DES COULEURS", "DES VISAGES", "ALE
XIE AGNOSIQUE", "TEXTUELLE", "COPIE ECHOUÉE", "SANS OBJET", "CASE VIDE"

150 VTAB 10: INVERSE : PRINT " "

151 PRINT "BILAPH "

152 PRINT TAB(8)"REDACTION D'UN BILAN D'APHASIE"

153 NORMAL

157 PRINT " "

158 PRINT "***** PAR A DELAFOSSE"

159 PRINT " SEPT 86"

160 PRINT " "

181 PRINT "NOM PRENOM AGE";

182 INVERSE : PRINT "VIRGULE";

183 NORMAL : PRINT "SERVICE DATE"

```

185 INPUT NO$,SE$
189 PRINT " "
190 PRINT "PARTICULARITES"
200 INPUT PA$
201 HOME
280 N = 0:D = 0
284 FOR I = 1 TO 9
285 PRINT A$(0,I);" "; CHR$( I + 64);"/";
286 NEXT I
290 N = N + 1
300 IF N > 20 GOTO 739
305 PRINT " "
306 PRINT " "
310 PRINT A$(N,1)
320 K$ = A$(N,1)
330 D = D + 1
350 FOR I = 2 TO 9
360 PRINT TAB( 5)A$(N,I);" ";I
370 NEXT I
375 IF N > 8 AND N < 11 GOTO 505
379 INVERSE
380 PRINT "TAPE SANS ESPACE NI VIRGULE LES LETTRES ET CHIFFRES CHOISIS"
390 INPUT J$
459 F$ = " "
460 FOR I = 1 TO LEN (J$)
461 IF ASC ( MID$( J$,I,1)) < 65 GOTO 470
462 F$ = F$ + A$(0,( ASC ( MID$( J$,I,1))) - 64) + " "
463 GOTO 480
470 F$ = F$ + A$(N, VAL ( MID$( J$,I,1))) + " "
480 NEXT I
490 IF F$ = " " CASE VIDE " GOTO 600
495 IF F$ = " SANS OBJET " THEN F$ = " "
497 G$(D) = K$ + "....." + F$
498 INVERSE : PRINT G$(D);" ":"EST EXACT O/N": NORMAL
499 INPUT B$
500 IF B$ = "N" THEN N = N - 1:D = D - 1: GOTO 502
501 IF F$ = " " THEN G$(D) = " "
502 HOME
503 IF N > 7 AND N < 10 GOTO 290
504 GOTO 284
505 INVERSE : PRINT "TAPE AUTANT DE P QUE DE REPONSES";
506 INPUT B$
507 PRINT "TAPE LE NUMERO DE REPONSE";
508 NORMAL : PRINT "VIRGULE";
509 INVERSE : PRINT "LE NOMBRE": NORMAL
510 FOR I = 1 TO LEN (B$)
515 INPUT J(I),O(I)
520 NEXT I
525 F$ = " "

```

```

530 FOR I = 1 TO LEN (B$)
540 F$ = F$ + A$(N,J(I)) + " " + STR$(O(I)) + "/"
541 NEXT I
545 PRINT "AJOUTE CE QUE TU VEUX OU N"
548 NORMAL
550 INPUT AJ$
552 IF AJ$ = "N" GOTO 570
560 F$ = F$ + AJ$
570 GOTO 497
600 PRINT "TAPE CE QUE TU VEUX"
602 INPUT C$
604 F$ = C$
610 GOTO 497
739 HOME
740 PRINT "*** CONCLUSION EN 3 LIGNES ***"
741 FOR D = 21 TO 23
742 INPUT G$(D)
743 NEXT D
744 PRINT "NOM DE L'EXAMINATEUR":
745 INPUT SI$
747 PRINT "SI L'IMPRIMANTE EST PRETE : TAPE SUR P"
748 INPUT B$
749 IF ASC (B$) < > 80 GOTO 747
750 PR# 1
760 PRINT ""
770 PRINT TAB( 20)NO$:"ANS"
780 PRINT TAB( 20)"SERVICE": " ";SE$
785 PRINT TAB( 5)PA$
790 FOR D = 1 TO 23
791 L = LEN (G$(D))
792 IF L < 81 GOTO 800
794 FOR I = 80 TO 60 STEP - 1
795 IF MID$( G$(D),I,1) = " " GOTO 797
796 NEXT I
797 GA$(D) = LEFT$( G$(D),I)
798 RI$(D) = RIGHT$( G$(D),L - I)
799 GOTO 801
800 GA$(D) = G$(D):RI$(D) = " "
801 IF D = 21 THEN PRINT "*** CONCLUSION ***"
802 PRINT GA$(D)
803 IF RI$(D) = " " GOTO 810
804 PRINT TAB( 10)RI$(D)
810 NEXT D
860 PRINT TAB( 56)SI$
870 PR# 0
880 PRINT "UN AUTRE EXEMPLAIRE O/N"
890 INPUT B$
900 IF B$ = "O" GOTO 747

```


Commentaires

- 10 crée un tableau A\$ de 20 lignes sur 9 colonnes représentant les 20 épreuves, leur titre et les 8 options de 2 à 9 (la ligne zéro propose 6 mots passe-partout et 3 signes de ponctuation : le point-virgule et les parenthèses, utilisables dans toutes les épreuves)
- Un tableau de 23 chaînes G\$ représentant les réponses obtenues aux 20 épreuves et les 3 lignes de conclusion (ces 23 chaînes seront scindées en lambeaux de gauche GA\$ et de droite RI\$ lors de l'impression).
- 20 à 55 lecture du tableau A\$
- 60 DATA : la ligne zéro
- 61 à 80 DATA : énoncé des 20 épreuves et, pour chacune, les 8 options possibles.
- 181 à 185 entrée des NOM PRENOM AGE du patient, NOM DU SERVICE DATE de l'examen
- 190 à 200 ligne PARTICULARITES réservée à tout caractère remarquable (bilinguisme, analphabétisme, dysarthrie ou pathologie ORL associée). Si aucune particularité, appuyer simplement sur la barre d'espacement.
- 280 N comptera les épreuves proposées. D comptera les réponses obtenues.
- 284 à 286 la ligne zéro du tableau (6 mots passe-partout et 3 signes de ponctuation, codés par 9 lettres de A à I) restera affichée en haut de l'écran.
- 290 à 300 après les 20 épreuves, les conclusions seront demandées (739).
- 310 titre de l'épreuve.
- 350 à 370 affichage des 8 options possibles, pour cette épreuve (codées de 2 à 9).
- 375 cas particulier des 9e et 10e lignes (dénomination : voir 505 à 570)
- 390 saisie au clavier de la ou des options à l'épreuve affichée, sous forme d'une suite de chiffres et-ou de lettres NON ESPACES.
- 460 à 480 la suite de lettres (462) et-ou de chiffres (470) est décodée selon leur place dans le tableau A\$.
- 490 si aucune option n'était satisfaisante, taper le 9 (CASE VIDE) et la réponse adéquate pourra être entrée en toutes lettres (600).
- 495 si l'épreuve n'était pas justifiée, taper le 8 (SANS OBJET) et cette rubrique n'apparaîtra pas à l'impression (501).
- 498 sécurité : la réponse est affichée en toutes lettres, si c'est exact taper la lettre O ; s'il y a eu erreur de frappe, taper N et la même épreuve sera à nouveau proposée (500).
- 504 et l'on repart pour l'épreuve suivante (284).
- 503 cas particulier de la dénomination, la ligne zéro ne sera pas affichée.
- 505 le P a été choisi pour sa proximité de RETURN sur le clavier (taper un ou plusieurs P à la suite, NON ESPACES).
- 507 à 515 entrer au clavier le numéro de la réponse suivi d'une virgule et du nombre de fois que cette réponse fut obtenue.
- 530 à 541 décodage selon le tableau A\$.
- 545 à 560 ligne libre pour ajouter au besoin un commentaire ; sinon taper N.
- 740 à 743 entrée au clavier des conclusions en 3 lignes (chacune peut occuper jusqu'à 150 caractères ou espaces ; les virgules, les

points et les guillemets sont interdits; les points-virgules et parenthèses autorisés).

- 744 et 745 nom de l'examineur.
- 747 à 749 sécurité au cas où l'imprimante n'aurait pas été préparée.
- 750 à 870 impression du bilan et des conclusions.
- 790 à 810 scission des lignes de plus de 80 caractères ou espaces, pour éviter de couper un mot lors du retour à la ligne.
- 880 à 900 possibilité d'imprimer plusieurs exemplaires.

Exemple



C... JACQUELINE 60 ANS Service Neurologique C PR PETIT 27 10 86

Licenciée en Lettres

Latéralisation Gauche préférentielle

Langage spontané réduit à des mots isolés ou rarement phrases de trois mots maximum.

..... bien articulé

..... syntaxe rudimentaire.

Langage automatique réussite sur comptage des jours; réussite partielle sur mois.

Répétition de mots anarthrie légère sur mots longs.

Répétition de phrases échec.

Praxies bucco faciales réussite sur imitation; réussite partielle sur ordre.

Dénomination réussites 3/par facilitation 3/retards 3/sans anarthrie
..... paraphasies 0/

Compréhension orale réussite aux 3 papiers à la 2e consigne.

Lecture à haute voix réussite sur substantifs.

Compréhension écrite réussite laborieuse aux mots-images.

Graphisme correct de la main gauche.

Orthographe paragraphies proches du mot-cible.

Signes spatiaux négligence droite dans l'appréhension de l'espace.

Conclusion

Chez une gauchère écrivant de la main droite, suppression totale du langage ayant régressé en 48 heures vers un tableau d'aphasie de Broca.

Indication d'une rééducation plurihebdomadaire.

Dr A. DELAFOSSE

BILAN DE C... JACQUELINE

Titre de l'épreuve	Entrée au clavier	Entrée(s) supplémentaire(s)
Latéralisation	35	réduit à des mots isolés ou rarement phrases de 3 mots maximum.
Langage spontané	9	
.....	2	
.....	4	
Langage automatique	A34GA25	
Répétition de mots	238	
Répétition de phrases	B	
Praxies bucco-faciales	A3GA24	
Dénomination	PPP	
.....	P	
Compréhension orale	A78	2,4 4,3 6,3 sans anarthrie
Lecture à haute voix	A5	2,0 N
Désignation écrite	8	
Compréhension écrite	A36	
Graphisme	28	
Orthographe	4	
Gerstmann	8	
Signes frontaux	8	
Signes spatiaux	24	
Agnosie visuelle	8	

Causes d'erreurs

— Ne pas confondre au clavier la lettre O et le zéro (0 barré) ni le l et le I majuscule.

— Ne pas taper de virgule dans une chaîne de texte, dans la ligne de particularités (2), ni dans la ligne libre (550), ni dans la suite de CASE VIDE (600), ni dans les 3 lignes de conclusion.

— En revanche, la virgule est obligatoire entre le NOM PRENOM AGE et le NOM DU SERVICE DATE ; de même elle est obligatoire, dans l'épreuve de dénomination entre chaque numéro et chaque nombre (507).

— Ne jamais taper ni point ni guillemets.

— Si l'on choisit le 9 (CASE VIDE), le 9 doit être tapé SEUL, aucun autre chiffre ou lettre ne doit l'accompagner ce qui empêcherait l'affichage de la rubrique TAPE CE QUE TU VEUX.

— De même, le 8 (SANS OBJET) doit être tapé SEUL pour entraîner la non-impression du titre de l'épreuve (501).

— La ligne DATA 70, soit la 10e ligne du tableau A\$, soit la 2e ligne de dénomination, ne comporte pas de SANS OBJET. Si elle était cependant sans objet, taper un seul P puis RETURN et 2 virgule zéro (la réponse affichée sera PARAPHASIES 0/)

— IMPORTANT : lors de la copie du LISTING, respecter l'espace qui existe après l'ouverture des guillemets et avec leur fermeture (lignes 490 et 495).

Modifications

— Il est possible d'ajouter des épreuves supplémentaires (description ; élaboration). Il faudra pour chaque épreuve prévoir 8 options possibles (soit 6 + SANS OBJET + CASE VIDE).

— L'ajout de 3 épreuves qualitatives demanderait les modifications suivantes :

```
9 REM dimensions des tableaux
10 DIM A$ (23,9), G$ (26), GA$ (26), RI$ (26)
20 FOR I = 1 TO 23
81 DATA
82 DATA
83 DATA
90 REM chaque DATA supplémentaire est suivie du nom de l'épreuve et
des 8 options possibles.
300 IF N > 23 GOTO 740
741 FOR D = 24 TO 26
790 FOR D = 1 TO 26
801 IF D = 24 THEN PRINT "CONCLUSION".
```

— Si les nouvelles épreuves sont quantitatives, modifier les lignes 375 et 503 selon le numéro de ligne des nouvelles épreuves.

Conclusion

— Ce programme ne peut se substituer au clinicien qui examine l'aphasique et tire les conclusions de l'examen. Il permet simplement d'accourcir le temps de rédaction d'un bilan. D'autres programmes sont disponibles pour les bilans des syndrômes frontaux et occipitaux. Critiques et suggestions seront les bienvenues.